

**IMPLEMENTASI ALGORITMA FUNDAMENTAL OSEANOGRAFI
(UNESCO 1983)**

MENGGUNAKAN BAHASA PEMROGRAMAN FORTRAN

LAPORAN

**Diajukan sebagai salah satu tugas mata kuliah Pemrograman dan
Oseanografi Komputasi pada Semester III Tahun Akademik 2025-2026**

Aisya Cahya Kamila (12924001)

Zahran Permana Hendrawan (12924032)

Zahra Syifa Alamsyah (12924062)



PROGRAM STUDI OSEANOGRAFI KOMPUTASI

INSTITUT TEKNOLOGI BANDUNG

BANDUNG

2025

BAB I

PENDAHULUAN

1.1. Latar Belakang

Kualitas analisis dalam oseanografi fisis sangat bergantung pada bagaimana data diolah. Parameter penting seperti densitas, salinitas, hingga suhu potensial harus dihitung menggunakan persamaan matematis, bukan sekadar hasil bacaan alat. Standar emas untuk perhitungan ini adalah algoritma UNESCO No. 44 (1983) yang berbasis PSS-78 dan EOS-80, yang hingga kini masih menjadi rujukan utama.

Masalahnya, pengolahan data terutama di kelas atau riset sederhana kerap kali masih manual. Padahal, rumus-rumusnya melibatkan deret polinomial panjang yang "menjebak". Risiko salah ketik atau salah hitung sangat besar jika dilakukan tanpa bantuan komputer. Kesalahan sekecil apa pun di sini bisa menurunkan kualitas data dan mengubah kesimpulan penelitian.

Oleh karena itu, otomasi menggunakan pemrograman sangatlah krusial. Fortran 90 adalah pilihan tepat karena bahasa ini dikenal stabil untuk hitungan sains dan memang digunakan dalam penyusunan algoritma asli UNESCO tersebut. Dengan membuat program berbasis Fortran, proses hitung menjadi jauh lebih cepat dan rapi. Yang terpenting, metode ini meminimalisir kesalahan manual, memastikan data yang dihasilkan akurat dan valid secara ilmiah.

1.2. Tujuan

- 1.2.1. Mengimplementasikan algoritma dari Paper UNESCO No. 44 (1983) sebagai acuan valid dalam pengolahan data.
- 1.2.2. Mengembangkan program komputasi berbasis Fortran 90 untuk mempercepat proses pengolahan data.
- 1.2.3. Menciptakan alat bantu hitung yang menghasilkan data presisi dan bebas dari kesalahan perhitungan manual.

BAB II

TINJAUAN PUSTAKA

2.1 Standarisasi Algoritma dalam Oseanografi Fisis

Dalam penelitian kelautan, konsistensi data adalah segalanya. Karena parameter fisik air laut saling mempengaruhi satu sama lain, penggunaan rumus yang berbeda sedikit saja bisa menghasilkan kesimpulan yang melenceng. Menyadari hal ini, UNESCO bekerja sama dengan panel ahli JPOTS (Joint Panel on Oceanographic Tables and Standards) merumuskan satu set algoritma baku. Tujuannya sederhana: agar siapa pun yang mengolah data, baik peneliti di Eropa maupun di Asia—mendapatkan hasil yang seragam jika input datanya sama. Algoritma ini dibangun di atas dua pilar utama, yaitu standar salinitas PSS-78 dan persamaan keadaan EOS-80. Standar ini menggantikan metode-metode lawas yang dinilai sudah tidak relevan dengan ketelitian alat ukur modern.

2.2 Skala Salinitas Praktis 1978 (PSS-78)

Berbeda dengan pendekatan lama yang mengandalkan klorinitas, PSS-78 mendefinisikan salinitas sebagai fungsi dari rasio konduktivitas listrik. Dalam algoritma ini, nilai salinitas ditentukan dengan membandingkan konduktivitas sampel air laut terhadap larutan standar Kalium Klorida (KCl).

Secara teknis, rasio konduktivitas (R) dipengaruhi oleh tiga variabel utama: salinitas (S), temperatur (t), dan tekanan (p). Algoritma PSS-78 memecah perhitungan ini menjadi tiga komponen koreksi:

1. Faktor tekanan (R_p)
2. Faktor temperatur (R_t)
3. Faktor temperatur referensi (r_t)

Dengan menggunakan persamaan polinomial, nilai konduktivitas yang terukur alat (CTD) dapat dikonversi menjadi salinitas praktis yang valid dalam rentang 2 hingga 42. Keunggulan lain dari algoritma ini adalah kemampuannya melakukan

perhitungan balik (*inverse*); yaitu mencari nilai konduktivitas dari nilai salinitas yang diketahui menggunakan metode iterasi numerik Newton-Raphson.

2.3 Persamaan Keadaan Internasional Air Laut (EOS-80)

Untuk urusan menghitung densitas (ρ) dan volume spesifik (V), dunia oseanografi kini berpegang pada EOS-80. Rumus ini dianggap jauh lebih reliabel ketimbang rumus Knudsen-Ekman yang dulu populer. Inti dari EOS-80 adalah penggunaan secant bulk modulus (K), sebuah fungsi yang memperhitungkan bagaimana air laut merespons tekanan, suhu, dan salinitas.

Ada satu trik komputasi menarik yang disarankan dalam dokumen UNESCO ini. Karena perubahan densitas di laut sebenarnya sangat tipis, komputer sering kali "kehilangan" angka penting di belakang koma jika kita menghitung menggunakan angka densitas penuh (seperti 1025 kg/m³). Solusinya, perhitungan dilakukan menggunakan Anomali Densitas (σ) atau Anomali Volume Spesifik (δ). Artinya, kita hanya menghitung selisihnya saja terhadap air laut standar. Dengan begini, ketelitian data bisa dijaga sampai detail yang sangat kecil (orde 10⁻⁸).

2.4 Parameter Fisis Turunan

Selain salinitas dan densitas, algoritma UNESCO No. 44 juga mencakup perhitungan parameter turunan yang krusial untuk analisis dinamika laut:

2.4.1 Konversi Tekanan ke Kedalaman

Seringkali alat ukur hanya mencatat tekanan (desibar), bukan kedalaman asli (meter). Konversinya tidak bisa sembarangan karena gravitasi bumi itu tidak rata; ia lebih kuat di kutub daripada di khatulistiwa. Algoritma standar ini menggunakan prinsip hidrostatik yang sudah dikoreksi dengan faktor lintang geografis dan densitas air, sehingga estimasi kedalaman menjadi sangat presisi.

2.4.2 Titik Beku (Freezing Point)

Air laut memiliki titik beku yang lebih rendah daripada air tawar. Temperatur titik beku (tf) ini akan semakin turun jika salinitas atau tekanan

meningkat. Millero dan Leung merumuskan persamaan polinomial untuk menghitung fenomena ini, yang sangat berguna dalam kajian oseanografi di daerah kutub atau laut dalam.

2.4.3 Panas Jenis (Specific Heat)

Berapa energi yang dibutuhkan untuk menaikkan suhu air laut? Jawabannya ada pada panas jenis (C_p). Nilai ini cukup dinamis: ia akan naik kalau suhu air naik, tapi akan turun jika airnya makin asin atau tekanannya makin besar. Algoritma menghitungnya secara bertahap, dimulai dari kondisi di permukaan laut, baru kemudian dikoreksi dengan faktor tekanan untuk kedalaman tertentu.

2.4.4 Gradien Adiabatik dan Temperatur Potensial

Saat massa air bergerak naik atau turun, tekanannya berubah drastis, dan ini memicu perubahan suhu alami tanpa ada panas yang masuk atau keluar (adiabatik).

- **Gradien Adiabatik (Γ):** Mengukur seberapa cepat suhu berubah akibat tekanan tersebut.
- **Temperatur Potensial (θ):** Ini adalah suhu "murni" dari parsel air seandainya ia dibawa ke permukaan laut tanpa pertukaran panas. Parameter ini sangat disukai peneliti untuk melacak pergerakan massa air karena nilainya stabil dan tidak terdistorsi oleh efek tekanan. Perhitungannya menggunakan metode numerik *Runge-Kutta* agar hasilnya presisi.

2.4.5 Kecepatan Suara (Sound Speed)

Kecepatan rambat bunyi di laut (U) dihitung menggunakan persamaan Chen dan Millero. Persamaan ini menggantikan rumus lama Wilson karena lebih konsisten dengan skala salinitas PSS-78. Kecepatan suara merupakan fungsi kompleks dari suhu, salinitas, dan tekanan, yang sangat vital untuk aplikasi akustik bawah air.

BAB III

METODOLOGI

3.1 Pendekatan Komputasi

Fokus utama dalam pengerjaan tugas ini adalah mengubah deretan rumus matematis dari dokumen *UNESCO Technical Papers in Marine Science No. 44 (1983)* menjadi barisan kode yang bisa dieksekusi. Kendala terbesarnya bukan sekadar menyalin rumus, tapi menjaga agar angka yang keluar tetap presisi. Rumus-rumus ini melibatkan koefisien yang sangat panjang—bisa mencapai belasan digit di belakang koma—sehingga risiko hasil melenceng akibat pembulatan otomatis komputer (*rounding error*) sangat tinggi jika sembarangan memilih alat bantu.

Atas dasar itulah Fortran 90 dipilih sebagai fondasi program. Pemilihan ini bukan cuma soal tradisi, tapi karena arsitektur Fortran memang dibangun spesifik untuk menangani komputasi sains yang berat. Algoritma asli UNESCO pun dikembangkan dengan logika yang selaras dengan bahasa ini, sehingga "terjemahan" ke dalam kodenya menjadi lebih akurat dibandingkan menggunakan bahasa pemrograman umum lainnya.

3.2 Konstruksi Algoritma

Program dibangun dengan memecah masalah besar menjadi beberapa sub-modul yang lebih kecil. Masing-masing modul bertanggung jawab untuk menghitung satu parameter spesifik sesuai referensi UNESCO:

1. **Modul Salinitas (SAL78):** Ini adalah gerbang utama pengolahan data. Modul ini dirancang fleksibel; ia bisa mengubah data mentah konduktivitas menjadi salinitas, atau sebaliknya, melakukan perhitungan mundur (*inverse*) untuk mencari konduktivitas jika nilai salinitasnya yang diketahui.
2. **Modul Densitas dan Volume (SVAN):** Menggunakan persamaan EOS-80, bagian ini menghitung seberapa padat air laut pada kondisi tertentu. Perhitungan tidak dilakukan pada nilai mutlak, melainkan pada nilai anomalnya, untuk menjaga ketelitian angka desimal.

3. **Modul Kedalaman (DEPTH):** Karena tekanan air (desibar) tidak berbanding lurus 1:1 dengan kedalaman (meter), modul ini menghitung konversinya dengan mempertimbangkan fakta bahwa gravitasi bumi berbeda-beda tergantung posisi lintang kita.
4. **Modul Termodinamika:** Bagian ini menangani parameter-parameter yang berkaitan dengan suhu dan energi, seperti titik beku air laut (*TF*), seberapa banyak panas yang bisa disimpan air (*CPSW*), hingga bagaimana suhu berubah secara alami akibat tekanan (*ATG* dan *THETA*).
5. **Modul Akustik (SVEL):** Bagian terakhir yang menghitung seberapa cepat suara merambat di dalam air, yang sangat dipengaruhi oleh kombinasi suhu, tekanan, dan kadar garam.


3.2 Flowchart

Agar program ini tidak sekedar menjadi kalkulator kaku, alur logikanya dirancang interaktif. Langkah demi langkahnya ialah :

Langkah pertama adalah **input variabel lingkungan dasar**. Program akan meminta data suhu dan tekanan terlebih dahulu karena kedua data ini hampir selalu ada dalam pengukuran oseanografi apa pun.

Selanjutnya, program menerapkan **logika seleksi input**. Karena data lapangan bisa bervariasi, kadang kita punya data konduktivitas dari alat CTD, kadang kita sudah punya data salinitas matang, program memberikan pilihan "Mode". Jika pengguna memilih Mode 0, program akan meminta input konduktivitas lalu menghitung salinitasnya. Jika memilih Mode 1, pengguna bisa langsung memasukkan angka salinitas.

Setelah nilai salinitas didapatkan (entah dari input langsung atau hasil hitungan), program secara otomatis **mengeksekusi seluruh fungsi turunan**. Namun, ada beberapa parameter spesial seperti Kedalaman dan Temperatur Potensial yang butuh data ekstra. Di tahap ini, program akan kembali bertanya kepada pengguna untuk memasukkan data Lintang atau Tekanan Referensi yang diinginkan. Hasil akhirnya kemudian disajikan secara berurutan di layar terminal.



Gambar 3.1 Flowchart AIO_OCEANOGRAPHY

BAB IV

HASIL DAN PEMBAHASAN

4.1. Hasil

Berikut hasil dari perhitungan dalam program yang telah dibuat:

```
=== KALKULATOR OSEANOGRAFI (UNESCO 44) ===

PILIH MODE INPUT:
0 = Input via KONDUKTIVITAS (Hitung Salinitas)
1 = Input via SALINITAS (Hitung Konduktivitas)
1

INPUT: Temperatur (C):
30
INPUT: Tekanan (Decibars):
1000
INPUT: Salinitas (PSS-78):
25
>> HASIL KONDUKTIVITAS (R)      :    1.01395154

--- PARAMETER TURUNAN ---
1. Freezing Point (TF)          :   -2.11137176
2. Sound Speed (SVEL)           :   1622.33447
3. Specific Heat (CPSW)          :   4026.92529
4. Adiabatic Lapse Rate (ATG)    :    2.46717042E-04
5. Spec. Vol. Anomaly (SVAN)     :   1358.26025
   Density Anomaly (SIGMA)       :   18.5300655

--- CALCULATIONS REQUIRING EXTRA INPUT ---
INPUT: Lintang (Degrees) for DEPTH:
45
>> Depth (M)                    :    989.499939

INPUT: Tekanan Referensi (db) for THETA:
1000
>> Potential Temp (THETA)       :    30.0000000
```

Gambar 4.1 Hasil Perhitungan Program

Berdasarkan perhitungan menggunakan program kalkulator oseanografi berbasis *UNESCO No. 44*, dengan masukan suhu 30 °C, tekanan 1000 dbar, dan salinitas 25 (PSS-78), diperoleh rasio konduktivitas (R) sebesar b. Parameter turunan yang dihasilkan meliputi suhu titik beku (TF) sebesar -2,11 °C, kecepatan rambat bunyi (SVEL) sebesar 1622,33 m/s, kalor jenis air laut (CPSW) sebesar 4026,93 J kg⁻¹ °C⁻¹, serta laju adiabatik (ATG) sebesar $2,47 \times 10^{-4}$ °C/dbar.

Selain itu, nilai *specific volume anomaly* (SVAN) yang diperoleh adalah $1358,26 \times 10^{-8}$ m³/kg, dan *density anomaly* (SIGMA) sebesar 18,53 kg/m³. Untuk perhitungan tambahan, dengan lintang 45°, tekanan 1000 dbar menghasilkan kedalaman sekitar 989,5 m, sedangkan suhu potensial (THETA) pada tekanan referensi 1000 dbar bernilai 30 °C.

4.2. Pembahasan

PRESSURE DECIBARS	CONDUCTIVITY RATIO R TEMPERATURE °C IPTS-68					SALINITY: 25
	0	10	20	30	40	
0	0.498008	0.654990	0.825481	1.006639	1.195796	
1000	0.506244	0.662975	0.833139	1.013951	1.202767	
2000	0.513925	0.670435	0.840302	1.020797	1.209296	
3000	0.521068	0.677385	0.846983	1.027187	1.215394	

Gambar 4.2.1 Tabel Rasio Konduktivitas terhadap Tekanan, Salinitas, dan Temperatur

Nilai rasio konduktivitas (R) yang diperoleh menunjukkan kesesuaian dengan tabel konversi salinitas–konduktivitas pada *UNESCO No. 44*, di mana untuk salinitas menengah (sekitar 25) pada tekanan tinggi, nilai R berada sedikit di atas 1. Hal ini mengindikasikan bahwa algoritma konversi salinitas telah diimplementasikan dengan benar dan konsisten dengan standar PSS-78.

PRESSURE DECIBARS	SPECIFIC HEAT SEAWATER C_p [J/(KG°C)] TEMPERATURE °C IPTS-68					SALINITY: 25
	0	10	20	30	40	
0	4048.4	4041.8	4034.8	4029.1	4051.2	
1000	4011.5	4012.9	4020.2	4026.9	4031.8	
2000	4070.0	4086.3	4097.4	4096.2	4093.6	

Gambar 4.2.2 Tabel Panas Spesifik Air Laut

Kalor jenis air laut yang berada pada $4026.9 \text{ J kg}^{-1} \text{ °C}^{-1}$ sesuai dengan tabel *specific heat of seawater* dalam paper UNESCO, yang menunjukkan variasi kalor jenis terhadap kondisi suhu, salinitas, dan tekanan. Nilai laju adiabatik yang relatif kecil mencerminkan perubahan suhu adiabatik yang terbatas terhadap tekanan, sebagaimana ditunjukkan dalam tabel adiabatik pada paper.

TABLE OF DEPTH (METERS)					
PRESSURE DECIBARS	LATITUDE (DEGREES)				
	0	30	45	60	90
500	496.65	496.00	495.34	494.69	494.03
1000	992.12	990.81	989.50	988.19	986.88
2000	1979.55	1976.94	1974.33	1971.72	1969.11
3000	2962.43	2958.52	2954.61	2950.71	2946.81

Gambar 4.2.3 Tabel Kedalaman

Hasil konversi tekanan ke kedalaman sebesar $\pm 989,5 \text{ m}$ pada lintang 45° juga konsisten dengan tabel tekanan–kedalaman UNESCO, yang menunjukkan bahwa tekanan 1000 dbar berkorelasi dengan kedalaman mendekati 1000 m di lintang menengah.

Secara keseluruhan, kesesuaian hasil komputasi dengan nilai dan kisaran pada tabel *UNESCO Technical Papers in Marine Science No. 44*

(1983) menegaskan bahwa program Fortran yang dikembangkan mampu merepresentasikan proses fisis laut secara akurat dan dapat digunakan sebagai alat bantu pengolahan data oseanografi.

BAB V

KESIMPULAN DAN SARAN

5.1 Kesimpulan

Berdasarkan hasil yang telah diperoleh, maka dapat disimpulkan beberapa hal.

1. Algoritma yang dibangun berhasil menghasilkan *output* yang konsisten dengan standar referensi UNESCO NO. 44 tahun 1983.
2. Program berbasis Fortran 90 mampu mempercepat perhitungan parameter oseanografi secara terintegrasi dibandingkan dengan metode manual.
3. Hasil validasi data (*check values*) menunjukkan tingkat akurasi yang tinggi dan bebas dari *human error* sesuai dengan standar referensi UNESCO.

5.2 Saran

Pada umumnya, data oseanografi di lapangan seperti data CTD atau ACDP berjumlah ratusan hingga ribuan baris. Keterbatasan fitur input data pada program ini menjadikan program kurang efisien dan rentan kesalahan ketik karena data diharuskan diinput satu per satu. Oleh karena itu, untuk pengembangan selanjutnya terkait program ini disarankan agar menambahkan fitur membaca data dari file eksternal seperti pada format *.csv* atau *.txt*. untuk memfasilitasi pengolahan data lapangan dalam jumlah besar secara efisien.

LAMPIRAN

```

PROGRAM KELOMPOK_4
  IMPLICIT NONE

  REAL :: S_IN, T_IN, P_IN, LAT_IN, PR_IN, CND_IN
  INTEGER :: M_IN
  REAL :: V_OUT_CND, V_OUT_SVAN, V_OUT_DEPTH, V_OUT_TF
  REAL :: V_OUT_CPSW, V_OUT_ATG, V_OUT_THETA, V_OUT_SVEL
  REAL :: SIGMA_RES

  REAL, EXTERNAL :: SAL78, SVAN, DEPTH, TF, CPSW, ATG, THETA, SVEL

  PRINT *, "=== KALKULATOR OSEANOGRAFI (UNESCO 44) ==="
  PRINT *, ""

  ! 1. PILIH MODE INPUT
  PRINT *, "PILIH MODE INPUT:"
  PRINT *, " 0 = Input via KONDUKTIVITAS (Hitung Salinitas)"
  PRINT *, " 1 = Input via SALINITAS (Hitung Konduktivitas)"
  READ *, M_IN
  PRINT *, ""

  ! 2. INPUT DATA UMUM
  PRINT *, "INPUT: Temperatur (C):"
  READ *, T_IN
  PRINT *, "INPUT: Tekanan (Decibars):"
  READ *, P_IN

  ! 3. LOGIKA SALINITAS/KONDUKTIVITAS
  IF (M_IN.EQ. 0) THEN
    PRINT *, "INPUT: Rasio Konduktivitas (R):"
    READ *, CND_IN
    ! Hitung Salinitas dari Konduktivitas
    S_IN = SAL78(CND_IN, T_IN, P_IN, 0)
    PRINT *, ">> HASIL SALINITAS (PSS-78) : ", S_IN
  ELSE
    PRINT *, "INPUT: Salinitas (PSS-78):"
    READ *, S_IN
    ! Hitung Konduktivitas dari Salinitas
    V_OUT_CND = SAL78(S_IN, T_IN, P_IN, 1)
    PRINT *, ">> HASIL KONDUKTIVITAS (R) : ", V_OUT_CND
  END IF
  PRINT *, ""

  ! 4. HITUNG PARAMETER LAIN (MENGUNAKAN S_IN)
  PRINT *, "---- PARAMETER TURUNAN ----"

  V_OUT_TF = TF(S_IN, P_IN)
  PRINT *, "1. Freezing Point (TF) : ", V_OUT_TF

  V_OUT_SVEL = SVEL(S_IN, T_IN, P_IN)
  PRINT *, "2. Sound Speed (SVEL) : ", V_OUT_SVEL

  V_OUT_CPSW = CPSW(S_IN, T_IN, P_IN)
  PRINT *, "3. Specific Heat (CPSW) : ", V_OUT_CPSW

  V_OUT_ATG = ATG(S_IN, T_IN, P_IN)
  PRINT *, "4. Adiabatic Lapse Rate (ATG) : ", V_OUT_ATG

  V_OUT_SVAN = SVAN(S_IN, T_IN, P_IN, SIGMA_RES)
  PRINT *, "5. Spec. Vol. Anomaly (SVAN) : ", V_OUT_SVAN
  PRINT *, " Density Anomaly (SIGMA) : ", SIGMA_RES
  PRINT *, ""

  ! 5. PARAMETER DENGAN INPUT TAMBAHAN
  PRINT *, "---- CALCULATIONS REQUIRING EXTRA INPUT ----"

  PRINT *, "INPUT: Lintang (Degrees) for DEPTH:"
  READ *, LAT_IN
  V_OUT_DEPTH = DEPTH(P_IN, LAT_IN)
  PRINT *, ">> Depth (M) : ", V_OUT_DEPTH
  PRINT *, ""

  PRINT *, "INPUT: Tekanan Referensi (db) for THETA:"
  READ *, PR_IN
  V_OUT_THETA = THETA(S_IN, T_IN, P_IN, PR_IN)
  PRINT *, ">> Potential Temp (THETA) : ", V_OUT_THETA
  PRINT *, ""

```

```

PRINT *, "Selesai. Tekan Enter untuk keluar..."
READ *, S_IN
END PROGRAM KELOMPOK_4

```

1. KONVERSI KONDUKTIVITAS <-> SALINITAS (SAL78)

```

REAL FUNCTION SAL78(CND,T,P,M)

  SAL(XR,XT) = (((((2.7081*XR-7.0261)*XR+14.0941)*XR+25.3851)*XR
& -0.1692)*XR + 0.0080
& + (XT/(1.0+0.0162*XT))*((( (-0.0144*XR+
& 0.0636)*XR-0.0375)*XR-0.0066)*XR-0.0056)*XR+0.0005)

  DSAL(XR,XT) = (((((13.5405*XR-28.1044)*XR+42.2823)*XR+50.7702)*XR
& -0.1692) + (XT/(1.0+0.0162*XT))*((( (-0.072*XR+0.2544)*XR
& -0.1125)*XR-0.0132)*XR-0.0056)

  RT35(XT) = (((1.0031E-9*XT-6.9698E-7)*XT+1.104259E-4)*XT
& + 2.00564E-2)*XT + 0.6766097

  C_POLY(XP) = ((3.989E-15*XP-6.370E-10)*XP+2.070E-5)*XP
  B_POLY(XT) = (4.464E-4*XT+3.426E-2)*XT + 1.0
  A_POLY(XT) = -3.107E-3*XT + 0.4215

  SAL78 = 0.0
  IF((M.EQ.0).AND.(CND.LE.5E-4)) RETURN
  IF((M.EQ.1).AND.(CND.LE.0.02)) RETURN

  DT = T - 15.0
  IF(M.EQ.1) GO TO 10

  R = CND
  RT = R/(RT35(T)*(1.0 + C_POLY(P)/(B_POLY(T) + A_POLY(T)*R)))
  RT = SQRT(ABS(RT))
  SAL78 = SAL(RT, DT)
  RETURN

10 RT = SQRT(CND/35.0)
  SI = SAL(RT,DT)
  N = 0

15 RT = RT + (CND - SI)/DSAL(RT,DT)
  SI = SAL(RT,DT)
  N = N + 1
  DELS = ABS(SI - CND)
  IF((DELS.GT.1.0E-4).AND.(N.LT.10)) GO TO 15

  RTT = RT35(T)*RT*RT
  AT = A_POLY(T)
  BT = B_POLY(T)
  CP = C_POLY(P)
  CP = RTT*(CP+BT)
  BT = BT - RTT*AT

  R = (SQRT(ABS(BT*BT + 4.0*AT*CP)) - BT)/(2.0*AT)
  SAL78 = R
  RETURN
END

```

3. ANOMALI VOLUME SPESIFIK & DENSITAS (SVAN)

```

REAL FUNCTION SVAN(S,T,P0,SIGMA)

  REAL P,T,S,SIG,SR,R1,R2,R3,R4
  REAL A,B,C,D,E,A1,B1,AW,BW,K,KO,KW,K35
  EQUIVALENCE (E,D,B1), (BW,B,R3), (C,A1,R2)
  EQUIVALENCE (AW,A,R1), (KW,KO,K)

  DATA R3500, R4/1028.1063, 4.8314E-4/
  DATA DR350/28.106331/

  P = P0/10.
  SR = SQRT(ABS(S))

  R1 = (((((6.536332E-9*T-1.120083E-6)*T+1.001685E-4)*T
& -9.095290E-3)*T+6.793952E-2)*T-28.263737

```

```

R2 = (((5.3875E-9*T-8.2467E-7)*T+7.6438E-5)*T-4.0899E-3)*T      &
& + 8.24493E-1
R3 = (-1.6546E-6*T+1.0227E-4)*T-5.72466E-3
SIG = (R4*S + R3*SR + R2)*S + R1

V350P = 1.0/R3500
SVA = -SIG*V350P/(R3500+SIG)
SIGMA = SIG+DR350

SVAN = SVA*1.0E+8
IF(P.EQ.0.0) RETURN

E = (9.1697E-10*T+2.0816E-8)*T-9.9348E-7
BW = (5.2787E-8*T-6.12293E-6)*T+3.47718E-5
B = BW + E*S

D = 1.91075E-4
C = (-1.6078E-6*T-1.0981E-5)*T+2.2838E-3
AW = ((-5.77905E-7*T+1.16092E-4)*T+1.43713E-3)*T - 0.1194975
A = AW + (C + D*SR)*S

B1 = (-5.3009E-4*T+1.6483E-2)*T+7.944E-2
A1 = ((-6.1670E-5*T+1.09987E-2)*T-0.603459)*T+54.6746
KW = ((-5.155288E-5*T+1.360477E-2)*T-2.327105)*T      &
& + 148.4206
KW = KW*T - 1930.06
KO = KW + (A1 + B1*SR)*S

DK = (B*P + A)*P + KO
K35 = (5.03217E-5*P+3.359406)*P+21582.27
GAM = P/K35
PK = 1.0 - GAM
SVA = SVA*PK + (V350P+SVA)*P*DK/(K35*(K35+DK))

SVAN = SVA*1.0E+8
V350P = V350P*PK

DR35P = GAM/V350P
DVAN = SVA/(V350P*(V350P+SVA))
SIGMA = DR350 + DR35P - DVAN
RETURN
END

```

4. KONVERSI TEKANAN KE KEDALAMAN (DEPTH)

```

REAL FUNCTION DEPTH(P, LAT)
REAL LAT
X = SIN(LAT/57.29578)
X = X*X

GR = 9.780318*(1.0+(5.2788E-3+2.36E-5*X)*X) + 1.092E-6*P
DEPTH = (((-1.82E-15*P+2.279E-10)*P-2.2512E-5)*P+9.72659)*P
DEPTH = DEPTH/GR
RETURN
END

```

5. TEMPERATUR TITIK BEKU (TF)

```

REAL FUNCTION TF(S,P)
TF = (-.0575+1.710523E-3*SQRT(ABS(S))-2.154996E-4*S)*S - 7.53E-4*P
RETURN
END

```

6. PANAS JENIS (CPSW)

```

REAL FUNCTION CPSW(S,T,P0)
P = P0/10.
SR = SQRT(ABS(S))

A = (-1.38385E-3*T+0.1072763)*T-7.643575
B = (5.148E-5*T-4.07718E-3)*T+0.1770383
C = (((2.093236E-5*T-2.654387E-3)*T+0.1412855)*T      &
& -3.720283)*T+4217.4

```

```

CP0 = (B*SR + A)*S + C

A = (((1.7168E-8*T+2.0357E-6)*T-3.13885E-4)*T+1.45747E-2)*T &
& -0.49592
B = (((2.2956E-11*T-4.0027E-9)*T+2.87533E-7)*T-1.08645E-5)*T &
& + 2.4931E-4
C = ((6.136E-13*T-6.5637E-11)*T+2.6380E-9)*T-5.422E-8
CP1 = ((C*P+B)*P+A)*P

A = (((-2.9179E-10*T+2.5941E-8)*T+9.802E-7)*T-1.28315E-4)*T &
& + 4.9247E-3
B = (3.122E-8*T-1.517E-6)*T-1.2331E-4
A = (A+B*SR)*S

B = ((1.8448E-11*T-2.3905E-9)*T+1.17054E-7)*T-2.9558E-6
B = (B+9.971E-8*SR)*S

C = (3.513E-13*T-1.7682E-11)*T+5.540E-10
C = (C-1.4300E-12*T*SR)*S
CP2 = ((C*P+B)*P+A)*P

CPSW = CP0 + CP1 + CP2
RETURN
END

```

7. ADIABATIC LAPSE RATE (ATG)

```

REAL FUNCTION ATG(S,T,P)
DS = S - 35.0
ATG = (((-2.1687E-16*T+1.8676E-14)*T-4.6206E-13)*P &
& + ((2.7759E-12*T-1.1351E-10)*DS+((-5.4481E-14*T &
& + 8.733E-12)*T-6.7795E-10)*T+1.8741E-8))*P &
& + (-4.2393E-8*T+1.8932E-6)*DS &
& + ((6.6228E-10*T-6.836E-8)*T+8.5258E-6)*T+3.5803E-5
RETURN
END

```

8. TEMPERATUR POTENSIAL (THETA)

```

REAL FUNCTION THETA(S,TO,PO,PR)
P = PO
T = TO
H = PR - P

XK = H*ATG(S,T,P)
T = T + 0.5*XK
Q = XK
P = P + 0.5*H

XK = H*ATG(S,T,P)
T = T + 0.29289322*(XK-Q)
Q = 0.58578644*XK + 0.121320344*Q

XK = H*ATG(S,T,P)
T = T + 1.707106781*(XK-Q)
Q = 3.0*XK - 4.121320344*Q
P = P + 0.5*H

XK = H*ATG(S,T,P)
THETA = T + (XK-2.0*Q)/6.0
RETURN
END

```

9. KECEPATAN SUARA (SVEL)

```

REAL FUNCTION SVEL(S,T,P0)
EQUIVALENCE (A0,B0,C0), (A1,B1,C1), (A2,C2), (A3,C3)

P = P0/10.
SR = SQRT(ABS(S))
D = 1.727E-3 - 7.9836E-6*P

B1 = 7.3637E-5 + 1.7945E-7*T
B0 = -1.922E-2 - 4.42E-5*T

B = B0 + B1*P

A3 = (-3.389E-13*T+6.649E-12)*T+1.100E-10
A2 = ((7.988E-12*T-1.6002E-10)*T+9.1041E-9)*T-3.9064E-7
A1 = (((-2.0122E-10*T+1.0507E-6)*T-6.4885E-8)*T-1.2580E-5)*T &
& + 9.4742E-5
A0 = (((-3.21E-8*T+2.006E-6)*T+7.164E-5)*T-1.262E-2)*T+1.389
A = ((A3*P+A2)*P+A1)*P+A0

C3 = (-2.3643E-12*T+3.8504E-10)*T-9.7729E-9
C2 = (((1.0405E-12*T-2.5335E-10)*T+2.5974E-8)*T-1.7107E-6)*T &
& + 3.1260E-5
C1 = (((-6.1185E-10*T+1.3621E-7)*T-8.1788E-6)*T+6.8982E-4)*T &
& + 0.153563
C0 = (((((3.1464E-9*T-1.47800E-6)*T+3.3420E-4)*T-5.80852E-2)*T &
& + 5.03711)*T+1402.388
C = ((C3*P+C2)*P+C1)*P+C0

SVEL = C + (A + B*SR + D*S)*S
RETURN
END

```